

```

=====
; geoGopherCon: connection editor
=====
.if          Pass1
              .noeqin
              .include  geoGopherSym
              .include  geoGopherMac
              .include  geoGopher.inc
              .eqin

.endif

=====
; constants for connection editor "dialog"
; note: 320-CON_WIDE must divide by 8 evenly (to position icons)
=====
TTL_HIGH    =      12          ;title height
FLD_HIGH    =      20          ;height of text field, with offset
NUM_FLD     =        3          ;number of text fields
CON_HIGH    =      (TTL_HIGH+8)+(FLD_HIGH*NUM_FLD)+24 ;icon height+8=24
CON_TOP     =      (200-CON_HIGH)/2
TXT_TOP     =      CON_TOP+TTL_HIGH+8;top of input text string
TXT_HIGH    =      12          ;height of input text area
TXT_LEFT    =      40
TXT_WIDE    =      144
CON_WIDE    =      TXT_LEFT+TXT_WIDE+8
CON_LEFT    =      (320-CON_WIDE)/2
LBL_LEFT    =      CON_LEFT+8
LBL_BOT     =      CON_TOP+28   ;baseline of prompt string
SAV_LEFT    =      CON_LEFT+8   ;icon position
DEL_LEFT    =      CON_LEFT+72
CNC_LEFT    =      CON_LEFT+136
ICON_TOP    =      CON_TOP+(TTL_HIGH+8)+(FLD_HIGH*NUM_FLD)
=====
; Save/Delete/Cancel "dialog" (edit connection)
=====
cfgEdit:    sec
            jsr      winDims
            lda      #1          ;solid (shadow box)
            jsr      SetPattern
            jsr      Rectangle
            clc
            jsr      winDims
            lda      #0          ;clear
            jsr      SetPattern
            jsr      Rectangle
            jsr      titleBar
            clc
            jsr      winDims
            lda      #ff         ;solid line
            jsr      FrameRectangle

```

```

;
=====
lda    tPort          ;show labels and text fields
bne    10$
lda    #'7'
sta    tPort
lda    #'0'
sta    tPort+1
lda    #0
sta    tPort+2
10$   LoadB    a0L,0          ;field counter
LoadB    a1L,LBL_BOT
LoadB    a1H,TXT_TOP
20$   jsr     showFld        ;shows label and draws box for text area
AddVB   #FLD_HIGH,a1L
AddVB   #FLD_HIGH,a1H
inc     a0L
lda     a0L
cmp     #NUM_FLD
bne     20$
LoadW   r0,conIcons
jsr     Dolcons
jsr     getInput
rts

```

```

;
=====
; Draw striped title bar at top of screen.
=====

```

```

titleBar:  lda    #9          ;horizontal stripes
jsr     SetPattern
LoadB   r2L,CON_TOP
LoadB   r2H,CON_TOP+TTL_HIGH
LoadW   r3,CON_LEFT
LoadW   r4,(CON_LEFT+CON_WIDE)-1
jsr     Rectangle
LoadW   r0,#conTitle
jsr     strWidth          ;returns string width in a0
LoadB   r1H,CON_TOP+8
; ((window width - string width) / 2) + window left
LoadW   r11,CON_WIDE
SubW    a0,r11
clc
ror     r11H
ror     r11L
lda     r11L
clc
adc     #[CON_LEFT
sta     r11L
lda     r11H
adc     #]CON_LEFT
sta     r11H
jsr     PutString
rts

```

```

=====
; Display a text prompt, the bounding box for its input, and the
; default text for the input.
;           pass:      a0L, item counter
;                   a1L, baseline of prompt string
;                   a1H, top of input text string
=====
showFld:    lda      a0L
           asl      a
           tay
           lda      lblAddr,y
           sta      r0L
           iny
           lda      lblAddr,y
           sta      r0H
           MoveB    a1L,r1H
           LoadW   r11,LBL_LEFT
           jsr      PutString
           lda      a1H
           sta      r2L
           clc
           adc      #(TXT_HIGH-1)
           sta      r2H
           LoadW   r3,CON_LEFT+TXT_LEFT
           LoadW   r4,(CON_LEFT+TXT_LEFT+TXT_WIDE)-1
           lda      #$ff          ;solid line
           jsr      FrameRectangle
           lda      a0L
           asl      a
           tay
           lda      txtAddr,y
           sta      r0L
           iny
           lda      txtAddr,y
           sta      r0H
           MoveB    a1L,r1H
           LoadW   r11,CON_LEFT+TXT_LEFT+3
           jsr      PutString
           rts

```

```

; =====
; Get the dimensions of the connection editor "dialog".
;           pass:      carry set to get shadow box dimensions
; =====
winDims:    LoadB    r2L,CON_TOP
            LoadB    r2H,(CON_TOP+CON_HIGH)-1
            LoadW    r3,CON_LEFT
            LoadW    r4,(CON_LEFT+CON_WIDE)-1
            bcc     10$
            AddVB   8,r2L
            AddVB   8,r2H
            AddVW   8,r3
            AddVW   8,r4
10$         rts
; =====
; Clear connection editor area.
; =====
clrCon:     lda     #0
            sta     tName      ;clear entry fields
            sta     tHost
            sta     tPort
            lda     #2         ;50% stipple
            jsr     SetPattern
            jsr     winDims
            AddVB   8,r2H     ;shadow box
            AddVW   8,r4
            jsr     Rectangle
            rts
; =====
; Loop through input fields. Mouse click will select field.
; =====
getInput:   LoadW    rightMargin,(CON_LEFT+TXT_LEFT+TXT_WIDE)-3
            LoadW    StringFaultVector,0
            php
            sei
            LoadW    otherPressVector,conMouse
            plp
            LoadB    getting,#$ff    ;enable input
reGet:     LoadB    a0L,0             ;loop counter
            LoadB    a0H,TXT_TOP+2
getStrs:   lda     a0L
mGetStrs:  asl     a
            tax
            lda     txtAddrs,x
            sta     r0L
            lda     txtAddrs+1,x
            sta     r0H
            LoadB    r1L,0           ;no string fault
            MoveB   a0H,r1H         ;y position
            ldx     a0L
            lda     txtLens,x
            sta     r2L             ;max. input length
5$        LoadW    r11,CON_LEFT+TXT_LEFT+3;x position
            LoadW    keyVector,gotInput
            jsr     GetString
            rts
gotInput:  lda     getting
            bne    10$
            rts
10$       AddVB   #FLD_HIGH,a0H
            inc   a0L
            lda   a0L
            cmp   #NUM_FLD
            bne  getStrs
            beq  reGet

```

```

; =====
conMouse:  LoadB    r2L,TXT_TOP
           LoadB    r2H,(TXT_TOP+TXT_HIGH)-1
           LoadW    r3,CON_LEFT+TXT_LEFT
           LoadW    r4,(CON_LEFT+TXT_LEFT+TXT_WIDE)-1
           ldx      #0
10$        jsr      IsMselnRegion
           tay      ;force flags to be set
           bne     30$
           inx
           cpx     #NUM_FLD
           bne     20$
           rts
20$        AddVB   #FLD_HIGH,r2L
           AddVB   #FLD_HIGH,r2H
           bne     10$
30$        txa
           pha
           lda     #TXT_TOP+2
40$        dex
           bmi     50$
           clc
           adc     #FLD_HIGH
           bne     40$
50$        sta     a0H
           pla
           sta     a0L
           jmp     mGetStrs

```

```

; =====
; icon handlers for connection editor "dialog"
; =====
doSave:      jsr      openCfg
             jsr      validate
             bcc     10$
             jmp     cfgEdit
10$          jsr      killText
             lda     cfgRec
             cmp     #$ff          ;new record?
             bne     40$          ;no, have record no.
             ldx     #2
20$          lda     fileHeader,x  ;VLIR index
             beq     30$          ;found an empty record
             inx
             inx
             bne     20$
             LoadW  errMsg,cfgFull ;configuration file full
             LoadW  RecoverVector,rstrStd
             LoadW  r0,errorDB
             jsr     DoDlgBox
             bra     60$
30$          txa
             clc
             ror     a
             tax
             dex
             txa
40$          jsr     PointRecord
             jsr     cfgToBuf
             LoadW  r2,cfgLen
             LoadW  r7,cfgBuf
             jsr     WriteRecord
             txa
             bne     50$
             jsr     UpdateRecordFile
             txa
             bne     50$
             jsr     CloseRecordFile
             txa
50$          beq     60$
             pha
             LoadW  a8,savErr
             pla
60$          jsr     showCode
             jsr     clrCon
             jsr     clrStat
             jmp     cfgStart

```

```

; =====
doDelete:    jsr    openCfg
             bcc    5$
             LoadW  errMsg,cfgErr
             LoadW  RecoverVector,rstrStd
             LoadW  r0,errorDB
             jsr    DoDlgBox
             bra    20$
5$           lda    cfgRec
             sta    curRecord
             jsr    DeleteRecord
             txa
             bne    10$
             jsr    UpdateRecordFile
             txa
             bne    10$
             jsr    CloseRecordFile
             txa
             beq    20$
10$          pha
             LoadW  a8,delErr
             pla
             jsr    showCode
20$          jsr    killText
             jsr    clrCon
             jsr    clrStat
             jmp    cfgStart
; =====
doCancel:    jsr    killText
             jsr    clrCon
             jsr    clrStat
             jmp    cfgStart
; =====
iconNop:     rts                                ;for disabled Delete icon

```

```

; =====
; Validate configuration entries.
;         return: carry flag set on error, clear otherwise
; =====
validate:   lda     tName
            bne     10$
            LoadW  errMsg,noConnNm
            bra     60$
10$         lda     tHost
            bne     20$
            LoadW  errMsg,noHostNm
            bra     60$
20$         lda     tPort
            bne     30$
            LoadW  errMsg,noPortNo
            bra     60$
30$         ldx     #0
40$         lda     tPort,x
            beq     70$
            cmp     #'0'
            bcc     50$
            cmp     #'9'+1
            bcs     50$
            inx
            bne     40$
50$         LoadW  errMsg,badPort
60$         LoadW  r0,errorDB
            LoadW  RecoverVector,rstrStd
            jsr    DoDigBox
            sec
            rts
70$         jsr    getCfgs      ;check for duplicate name
            lda     numItems
            beq     120$
            LoadW  a8,cfgNames
            LoadW  a9,tName
80$         ldx     #a8
            ldy     #a9
            jsr    CmpString
            bne     90$
            LoadW  errMsg,dupCfg
            bra     60$
90$         ldy     #0
100$        lda     (a8),y
            beq     110$
            iny
            bne     100$
110$        iny
            iny      ;past record no.
            tya
            clc
            adc     a8L
            sta     a8L
            lda     a8H
            adc     #0
            sta     a8H
            CmpW   a8,cfgEnd
            bcc     80$
120$        clc
            rts

```



```

=====
; Copy to/from configuration buffer
=====
cfgToBuf:   ldx     #0
            ldy     #0
10$         lda     tName,x
            sta     cfgBuf,y
            beq     20$
            inx
            iny
            bne     10$
20$         ldx     #0
            iny
30$         lda     tHost,x
            sta     cfgBuf,y
            beq     40$
            inx
            iny
            bne     30$
40$         ldx     #0
            iny
50$         lda     tPort,x
            sta     cfgBuf,y
            beq     60$
            inx
            iny
            bne     50$
60$         rts
;
bufToCfg:   ldx     #0
            ldy     #0
10$         lda     cfgBuf,y
            sta     tName,x
            beq     20$
            inx
            iny
            bne     10$
20$         ldx     #0
            iny
30$         lda     cfgBuf,y
            sta     tHost,x
            beq     40$
            inx
            iny
            bne     30$
40$         ldx     #0
            iny
50$         lda     cfgBuf,y
            sta     tPort,x
            beq     60$
            inx
            iny
            bne     50$
60$         rts

```

```

=====
; Open configuration file (create if not found).
;      pass:      nothing
;      return:    carry set on error, clear otherwise
;                on error, .A contains error code
=====
openCfg:  LoadW   r6,cfgName
          jsr     FindFile
          txa
          beq     30$
          cmp     #FILE_NOT_FOUND
          bne     40$
          LoadW  r9,cfgHead      ;not found, create
          LoadB  r10L,1
          jsr     SaveFile
          txa
          bne     40$
10$      LoadW  r0,cfgName      ;open newly-created file
          jsr     OpenRecordFile
          txa
          bne     40$
20$      jsr     AppendRecord   ;mark all records as empty
          txa
          beq     20$
          cpx     #OUT_OF_RECORDS
          bne     40$
          lda     #0             ;first record is network card type
          jsr     PointRecord
          LoadW  r2,1
          LoadW  r7,cfgType     ;just write the one byte
          jsr     WriteRecord
          jsr     UpdateRecordFile
          txa
          bne     40$
          clc
          rts
30$      lda     dirEntryBuf+OFF_GFILE_TYPE
          cmp     #APPL_DATA
          bne     50$
          LoadW  r9,dirEntryBuf
          jsr     GetFHdrInfo
          txa
          bne     40$
          ldx     #0
35$      lda     fileHeader+O_GHFNAME,x
          cmp     cfgPerm,x     ; permanent name string match?
          bne     50$
          inx
          cpx     #20
          bne     35$
          LoadW  r0,cfgName     ; OK then, open file
          jsr     OpenRecordFile
          txa
          bne     40$
          clc
          rts
40$      pha
          pha                ;fatal error, must have config file
          LoadW  a8,cfgErr
          pla
          jsr     showCode
          jmp     EnterDeskTop
50$      LoadW  errMsg,badCfg   ;invalid configuration file
          LoadW  r0,errorDB
          LoadW  RecoverVector, rstrDone
          jsr     DoDlgBox
          jmp     EnterDeskTop   ;fatal error

```

```

; =====
conTitle:      .byte      " gopher connection ",0
lName:        .byte      "name:",0
lHost:        .byte      "host:",0
lPort:        .byte      "port:",0
lblAddr:      .word      lName,lHost,lPort
tName:        .byte      0
              .block     30
tHost:        .byte      0
              .block     30
tPort:        .byte      "70",0,0,0,0
txtLens:      .byte      30,30,5
txtAddr:      .word      tName,tHost,tPort
badCfg:       .byte      "Invalid configuration file!",0
cfgFull:      .byte      "Configuration file full!",0
noConnNm:    .byte      "You have not entered a name.",0
noHostNm:    .byte      "You have not entered a host name.",0
noPortNo:    .byte      "You have not entered a port number.",0
badPort:     .byte      "Invalid port number.",0
dupCfg:      .byte      "Duplicate configuration name.",0
cfgsErr:     .byte      "reading configurations!",0
cfgErr:      .byte      "opening configuration file!",0
savErr:      .byte      "saving configuration!",0
delErr:      .byte      "deleting configuration!",0
getting:     .byte      0          ;accepting input?
cfgBuf:      .block     68          ;31+31+6
cfgRec:      .block     1
cfgLen      =      cfgRec-cfgBuf
cfgName:     .byte      "geoGopher.cfg",0
cfgType:     .byte      MOD_ULT    ;which network card
selCfg:      .byte      0          ;index into listbox
selIndex:    .byte      0          ;list index of selected item
cfgEnd:      .word      0          ;pointer to end of list



```

```

; =====
; configuration file header
; =====
cfgHead:    .word    cfgName
            .byte    3,21,$BF
            .byte    $3c,$00,$3c
            .byte    $42,$00,$42
            .byte    $99,$ff,$99
            .byte    $bf,$00,$fd
            .byte    $bc,$00,$3d
            .byte    $5b,$c3,$da
            .byte    $37,$27,$2c
            .byte    $17,$27,$28
            .byte    $24,$24,$24
            .byte    $23,$c3,$c4
            .byte    $20,$18,$04
            .byte    $20,$7e,$04
            .byte    $20,$99,$04
            .byte    $20,$99,$04
            .byte    $20,$99,$04
            .byte    $20,$7e,$04
            .byte    $20,$00,$04
            .byte    $20,$00,$04
            .byte    $20,$00,$04
            .byte    $20,$00,$04
            .byte    $20,$00,$04
            .byte    $80 | USR
            .byte    APPL_DATA
            .byte    VLIR
            .word    $0000        ;load address
            .word    $ffff        ;end address
            .word    0            ;init address
cfgPerm:    .byte    "geoGopherCfgV1.0",0,0,0,0
            .byte    "Cenbe",0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0
            .byte    "geoGopher V2.1",0,0,0,0
            .block    23
            .byte    "geoGopher configuration file",0




```

```

=====
; Create/Open/Quit dialog (open or edit connection)
=====
connDB:      .byte      DEF_DB_POS | 1
             .byte      DBTXTSTR,TXT_LN_X,16
             .word      txtOpt
             .byte      DBUSRICON,2,24
             .word      crtIcon
             .byte      DBTXTSTR,71,34
             .word      txtNew
             .byte      OPEN,2,48
             .byte      DBTXTSTR,71,58
             .word      txtExist
             .byte      DBUSRICON,2,72
             .word      quitIcon
             .byte      DBTXTSTR,71,82
             .word      txtQuit
             .byte      0
;
txtOpt:      .byte      "Please Select Option:",0
txtNew:      .byte      "new connection",0
txtExist:    .byte      "existing connection",0
txtQuit:     .byte      "to deskTop",0
;
crtIcon:     .word      icnCrt
             .byte      0,0           ;ignore position, use DB
             .byte      6,16         ;standard icon size
             .word      doCreate
;
quitIcon:    .word      icnQuit
             .byte      0,0           ;ignore position, use DB
             .byte      6,16         ;standard icon size
             .word      doExit
;
icnCrt:
             
icnQuit:
             

```

```

=====
; icons for configuration editor
=====
conIcons:    .byte    3
             .word    SAV_LEFT+24
             .byte    ICON_TOP+8
;
             .word    icnSave    ;in resident module
             .byte    SAV_LEFT/8,ICON_TOP
             .byte    6,16
             .word    doSave
;
dellcon:     .word    icnDel
             .byte    DEL_LEFT/8,ICON_TOP
             .byte    6,16
delDspch:    .word    doDelete
;
             .word    icnCncl
             .byte    CNC_LEFT/8,ICON_TOP
             .byte    6,16
             .word    doCancel
;
icnDel:
             
icnDelDs:
             
icnCncl:
             
;
=====
cfgNames:    .block    1024    ;connection name buffer

```